
INTRODUCTION 

The Garmin GPS-16-HVS GPS receiver (available from Sutron as part 5000-0175) and the GPS-17-HVS 
(5000-0176) have RS232 interfaces compatible with the Sutron 9210/Xpert.  An Xpert Basic program has 
been written with a scheduled routine (GARMINGPS) to read the time from the GPS unit and set the time 
in the 9210/Xpert.  The GARMINGPS routine is written to read a GPS connected to COM2 to read a 
NEMA GPGGA message at 4800 baud.  The factory default configuration for the Garmin GPS-16 HVS 
includes the GPGGA message at 4800 baud so no additional configuration is normally needed.  Other 
NEMA messages may be output; however, the BASIC program will ignore them and only accept the 
GPGGA message. 

Once the GARMINGPS.BAS program has been loaded in to the system, routine must be scheduled using 
Setup, Basic.  Sutron recommends that the program be scheduled to run every 6 hours.   This can be 
changed to any interval desired in order to conserve power and still maintain time. 

The GARMIN GPS is wired to COM2 as follows.   

Note: Sutron also offers a RJ45 to DB9 adapter that implements the required wiring. 

 

RJ45 DB9 Female 

1 PWR + 9 

2 PWR GND 5 

3 On/Off -- use a DIO to control this or connect to GND to remain on always 

4 Data In 3 

5 Data Out 2  

 

Note that the GARMIN GPS gets power (+12V) from the DB9 pin 9.   Be sure to open the 9210/Xpert and 
set the jumper appropriately, see power ref. at the end of this App note. 

Note also that the GARMIN GPS can be turned on and off using the RJ45 pin 3 black wire.  The code is 
written to turn on and off the GPS using DIO 4.  This is done to help conserve power since it is only 
necessary to read the GPS a few times a day. 

You can interact directly with the GARMIN GPS through the 9210 for troubleshooting using the new 
Passthru command available at the command line prompt.  Stop recording and then issue the command 
Passthru com2:4800.  

The GPGGA message is an ASCII message with values separated by commas.  The program uses the 
“WAITFOR” function to pass by all the other data that may be output from the GPS and wait for the 
GPGGA message.  The Input statement is then to read all the values out into the appropriate variables.  
The program then checks to see if the GPS unit has satellites in view and a valid time.  If not, it loops until 
this happens or the loop times out after 900 seconds. 

With a valid time, the code will only set the time if it is more than 1 second different from the 9210 time.  
Note that the UTC time is used, not local time.  Additional checks to the time can be added if desired. 

The format of the GPGGA message is 

        

$GPGGA,<1>,<2>,<3>,<4>,<5>,<6>,<7>,<8>,<9>,M,<10>,<11>,<12>,*hh<CR><LF> 

 

where: 

<1> UTC time of position fix, hhmmss format 

<2> Latitude, ddmm.mmmmm format (leading zeros will be transmitted) 

<3> Latitude hemisphere, N or S 

<4> Longitude, dddmm.mmmm format (leading zeros will be transmitted) 

<5> Longitude hemisphere, E or W 

<6> GPS quality indication, 0=fix not available, 1= non-differential GPS 

fix available, 2=differential GPS fix available, 6=estimated 

<7> Number of satellites in use, 00 to 12 (leading zeros will be 

transmitted 

<8> Horizontal dilution of precision, 0.5 to 99.9 

<9> Antenna height above/below mean sea level, -9999.9 to 99999.9 meters 

<10> Geoidal height, -999.9 to 9999.9 meters 

<11> Differential GPS data age 

<12> Differential Reference Station ID, 0000 to 1023. 

*hh checksum (XOR sum of all characters between $ and *, non-inclusive. 

 
The details of the $GPGGA and the complete operations the Garmin GPS-16 are provided in a 
separate manual available from Garmin. 
 
GARMINGPS.BAS 

 
Public Sub SCHED_GARMINGPS 

 ' 5/31/2012 D.K. | Update notes, fix Input statement 

 ' This subroutine can be scheduled to run 

 ' periodically (e.g. every hour) to read the GPS 

 ' and set the Xpert/9210 clock. 

 ' The routine expects the GPS to be connected to COM3 and that the 

 ' power pin is on DIO module 1, channel 4 

 ' If those need changed, change in USER SET Constants 

 ' Make sure to set the port power to +12Volts  

 ' The routine expects the data in  

 ' NEMA $GPGGA format (default for GRAMIN) 

 ' For troubleshooting, statusmsg may be uncommented  

 '------- USER SET CONSTANTS ----- 

 CONST GPSPort = "COM3:" 

 CONST GPSPwrMod = 1 ' DIO module 

 CONST GPSPwrCH = 4  ' DIO Channel 

 '------- GARMIN Constants ------- 

 Const BAUD = 4800 

 Const NOPARITY = 0 

 Const NOHANDSHAKE = 0 

 Const GPSTimeout = 900 'Timeout in seconds to get a valid GPS signal 

 '-------------------------------- 

 'Open port, check for abort (recording off) often 

 Port = FreeFile 

 If Abort Then Goto ErrorHandler 

 On Error Resume Next 

 ' StatusMsg "Opening GPS port" 

 Open GPSPort as Port nowait 

 If Err <> 0 Then 

  ErrorMsg "Failed to open com port, err " & Err 

  Goto ErrorHandler 

 End If 

 ' StatusMsg "GPS port opened" 

 SetPort Port, BAUD, NOPARITY, 8, 1, NOHANDSHAKE 

 StartTime = Time ' for GPS timeout calculations 

 TimeValid = false ' to get us into the loop 

 Digital GPSPwrMod, GPSPwrCH, -1 'turn on the GPS 

 '''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''' 

 ' Loop until GPStimeout exceeded or 

 ' we receive a valid time from the GPS 

 Do 

  Sleep 1 'Moved delay to top of loop 

  If Abort Then Goto ErrorHandler 

  'Timeout to receive any data, not the same as GPS timeout 

  SetTimeout Port,35  

  'Port opened and initialized 

  FlushInput Port ' get rid of data in the input buffer 

  if Err then ErrorMsg "FlushInput failed " & Err 

  WaitFor Port, "$GPGGA," ' waitfor message identifier 

  If Err then 

   ErrorMsg "Failed to get $GPGGA" 

   Goto ErrorHandler 

  End If 

  ' StatusMsg "Got $GPGGA," 

  'Init variables 

  MSGID = 0 

  UTC = 0 

  Latitude = 0.0 

  LatH = "" 

  Longitude = 0.0 

  LongH = "" 

  GPSQual = 0 

  NumSats = 0 

  HorDil = 0.0 

  AntAlt = 0.0 

  Meters = "" 

  AgeDiff = 0.0 

  DiffRef = 0 

  Asterisk = "" 

  Cksum = 0 

  Input Port, UTC, Latitude, LatH, Longitude, LongH, GPSQual,      

  NumSats, HorDil, AntAlt, Meters, AgeDiff, DiffRef, Asterisk,  

  Cksum 

  If Err Then 'Did not get data 

   ErrorMsg "Input error " & Err 

   Goto ErrorHandler 

  End If 

  ' See if data is valid, if so, we can exit the loop 

  ' StatusMsg "MSG " &UTC &" " &latitude &" " &gpsqual &" "   

  ' & numsats 

  If (NumSats > 0) AND (GPSQual >0) Then TimeValid = true 

 Loop Until TimeValid OR ((Time - StartTime)> GPSTimeout) 

 ' 

 '''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''''' 

 If ((Time - StartTime) > GPSTimeout) then 

  ErrorMsg "Timeout waiting for GPS -- NumSats: " & NumSats 

  & " GPSQUAL: " & GPSQual 

  goto ErrorHandler 

 End If 

 Hr = Int(UTC / 10000) 

 Min = Int((UTC - (Hr * 10000))/100) 

 Sec = UTC Mod 100 

 GPSTime=TimeSerial(Hr, Min, Sec) 

 If ((Time-GPSTime)>1) OR ((GPSTime-Time)>1) then 

  Time = GPSTime 'Here is where we actually change the system time 

  StatusMsg "Basic GarminGPS changed system time to " & GPSTime 

 Else 

  'No need to fill the system log when we don't set the time 

  'StatusMsg "Time too close " &GPSTime  

 End If 

 

 ErrorHandler: 

 Close Port 

 Digital GPSPwrMod, GPSPwrCH, 0 ' turn off GPS 

End Sub 

Power on Pin 9 Reference. 
 
9210B / Xpert2: 
 
The following table gives the jumper settings and their default settings. To change the default 

settings, open the Xlite and change these jumpers: 

 

 COM1 COM2 COM3 COM4 

Configuration 

Jumper 

connector 

J800 J900 J901 J1000 

Default 

configuration 

for Pin 9 

Ring in Ring in +5V Switched Ring in 

 

In the switched configuration, power is turned on when the port is opened, and turned off when 

the port is closed.   

Xpert2 w/ Four additional com ports: 
 

 
 
 
Xpert (Old Revision) 
COM2-4 

 

 Functio

n 

COM2 and 4 Default 

(COM 6-9 if 

installed) 

COM3 

Default 

J1, 2-3 

J2, 2-3 

+5V  X 

J1, 2-3 

J2, 1-2 

+12V   

J1, 1 to 2 

J2, doesn't matter 

Ring 

IN  
X  

 

 

 

 

 

J 1   J2 
PIN 1 

. . . . . . . . .  

    . . . . . 

. 

1 

2 

Xlite (Old Revision) 
 

  

Function 

J16 

COM2 

Default 

J13 

COM3 

Default 

Pin 2 to 4 +5V  X 

Pin 6 to 4 +12V   

Pin 3 to 4 Ring IN  X  

 


